EXAMPLE 8.10-1: An Energy Recovery Wheel

A building at the local zoo in Madison, Wisconsin houses primates, large cats, visitors and staff in four separate zones. The focus of this problem is on the zone that houses the primates. The total volume of the zone is V_{zone} = 2500 m^3. In order to maintain the health of the animals, as well as to control odors so that the zoo is a pleasant place for visitors, it is necessary to ventilate the zone at a minimum rate of ac = 2.5 air changes per hour all of the time (i.e., 24 hours per day, 7 days a week). The outdoor air that replaces the ventilated air must be conditioned to T_b = 20°C. (Internal generation from lights and equipment provides the remaining heating needs). The system is shown in Figure 1(a).



Figure 1: Ventilation and conditioning system (a) without energy recovery wheel, and (b) with energy recovery wheel.

Zoo personnel have found that the costs of heating (in the winter heating season) and cooling (in the summer cooling season) the outdoor air are substantial. Therefore, you have been asked to look at alternatives for cost savings. One possibility is the use of a rotary regenerator for recovering energy from the exhaust air and transferring it to the outside, ventilation air (an energy recovery wheel, see Figure 8-66); such a system is shown schematically in Figure 1(b).

During the heating system, the energy recovery wheel accepts heat from the warm building air leaving the zone and transfers it to the outside air, pre-heating the air in order to reduce the heating that must be provided by the building conditioning equipment. During the cooling season the opposite happens; the energy recovery wheel rejects heat to the (relatively) cool air leaving the building and accepts heat from the warm outdoor air, reducing the cooling that must be provided by the conditioning equipment. Therefore, the energy recovery wheel provides year-round savings and is particularly attractive in applications where large ventilation rates are required.

The energy recovery wheel being considered for this application is made of aluminum with density ρ_r = 2700 kg/m^3 and c_r = 900 J/kg-K. The packing is made up of triangular channels, as shown in Figure 2. The thickness of the aluminum separating adjacent rows of channels is th_b = 0.3 mm and the thickness of the aluminum struts that separate adjacent passages is th_s = 0.1 mm. The channels themselves are H_p = 2.5 mm high and have a half-width of W_p = 1.5 mm. The diameter of the wheel is D_r = 0.828 m and the length of the wheel is L = 0.203 m. The wheel rotates at N = 30 rev/min and the matrix spends half of its time exposed to outside air and the other half exposed to building exhaust air.



Figure 2: Structure of the energy recovery wheel.

The average outdoor temperature (T_o) for each month in Madison, WI is provided in Table 1.

Table 1: Monthly-average ambient temperatures in Madison, WI

Month	Temperature (°C)
Jan	-8
Feb	-7
Mar	-1
Apr	7
May	13
Jun	19
Jul	21
Aug	20
Sep	15
Oct	10
Nov.	2
Dec	-6

Notice that the cooling load in Madison, WI is not substantial because the monthly average ambient temperature barely exceeds the building temperature in the warmest month. Therefore, the energy recovery wheel in this climate will primarily result in a savings in the heating energy required. The cost of providing heating is hc = \$10/GJ; neglect any cost savings associated with cooling. There is an operating cost associated with the additional fan power required to force the air through the rotary regenerator. Assume that the fan efficiency is η_{fan} = 0.5 and that the cost of electricity is ec = $0.105/kW-hr.

a.) Estimate the annual cost savings that would result from installation of the rotary regenerator wheel.

The input information is entered in EES:

```
"EXAMPLE 8.10-1: An Energy Recovery Wheel"
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

"Inputs"
V_zone=2500 [m^3]                              "volume of the zone that houses the primates"
ac=2.5 [1/hr]*convert(1/hr,1/s)                "air changes required for ventilation"
T_b=converttemp(C,K,20[C])                     "conditioned air temperature"
rho_r=2700 [kg/m^3]                            "density of regenerator material"
c_r=900 [J/kg-K]                               "specific heat capacity of regenerator material"
th_b=0.3 [mm]*convert(mm,m)                    "thickness of base plate between rows of channels"
th_s=0.1 [mm]*convert(mm,m)                    "thickness of struts between adjacent channels"
H_p=2.5 [mm]*convert(mm,m)                     "height of a channel"
W_p=1.5 [mm]*convert(mm,m)                     "half-width of a channel"
D_r=0.828 [m]                                  "wheel diameter"
L=0.203 [m]                                    "wheel length"
N=30 [1/min]*convert(1/min,1/s)                "rotation rate"
hc=10 [$/GJ]*convert($/GJ,$/J)                 "cost of heating"
ec=0.105 [$/kW-hr]*convert($/kW-hr,$/J)        "cost of electricity"
eta_fan=0.5 [-]                                "fan efficiency"
```

Initially, the calculation will be carried out for the month of January; the result will subsequently be extended for an entire year. The number of days (N_{day}) and average outdoor temperature (T_o) for January are entered:

```
"Monthly conditions"
Month$='Jan'                                   "month"
N_days=31 [day]                                "number of days in the month"
T_o_C=-8 [C]                                   "monthly average outdoor air temperature in C"
T_o=converttemp(C,K,T_o_C)                     "monthly average outdoor air temperature"
```

The number of days in the month is used to determine the total operating time during the month (t_{month}):

```
t_month=N_days*convert(day,s)"time associated with the month"
```

The matrix geometry shown in Figure 2 is analyzed in order to determine the characteristics of the packing. The cross-sectional area of a single passage is:

$$A_{c,p} = W_p\, H_p$$

and the wetted perimeter of a passage is:

$$p_p = 2W_p + 2\sqrt{H_p^2 + W_p^2}$$

The surface area of a passage is:

$$A_{s,p} = L\, p_p$$

The characteristic radius of the passage is defined according to Eq. (8-387):

$$r_{char} = \frac{L\, A_{c,p}}{A_{s,p}}$$

```
"matrix geometry"
A_c_p=H_p*W_p                          "cross-sectional area of a single passage"
p_p=2*W_p+2*sqrt(W_p^2+H_p^2)          "perimeter of a single passage"
A_s_p=p_p*L                            "surface area for a single passage"
r_char=L*A_c_p/A_s_p                   "hydraulic diameter of a single passage"
```

The frontal area of the regenerator wheel is:

$$A_{fr} = \pi \frac{D_r^2}{4}$$

The number of passages is the ratio of the frontal area to the approximate cross-sectional area of each passage including the surrounding aluminum:

$$N_p = \frac{A_p}{A_{c,p} + th_s \sqrt{W_p^2 + H_p^2} + W_p\, th_b}$$

```
A_fr=pi*D_r^2/4                                    "frontal area of the regenerator"
N_p=A_fr/(A_c_p+sqrt(W_p^2+H_p^2)*th_s+W_p*th_b)   "number of passages"
```

The total surface area associated with the matrix is the product of the number of passages and the surface area per passage:

$$A_s = N_p\, A_{s,p}$$

and the total void volume (i.e., the volume of the fluid entrained in the regenerator) is the product of the number of passages and the volume of each passage:

$$V_f = N_p\, A_{c,p}\, L$$

```
A_s=N_p*A_s_p                                      "total surface area"
```

```
V_f=L*A_c_p*N_p                                          "total void volume"
```

The total volume of the aluminum in the regenerator is:

$$V_r = A_{fr}\,L - V_f$$

and the porosity of the bed is:

$$\phi = \frac{V_f}{A_{fr}\,L}$$

```
phi=V_f/(A_fr*L)                                         "porosity of regenerator wheel"
V_r=A_fr*L-V_f                                           "regenerator material volume"
```

The air properties (ρ_a, μ_a, c_a, and Pr_a) are evaluated at the average temperature in the regenerator:

$$\overline{T}_a = \frac{T_o + T_b}{2}$$

```
"air properties"
T_a_avg=(T_o+T_b)/2                                      "average air temperature"
rho_a=density(Air,T=T_a_avg,P=1 [atm]*convert(atm,Pa))   "density of air"
mu_a=viscosity(Air,T=T_a_avg)                            "viscosity of air"
c_a=cP(Air,T=T_a_avg)                                    "specific heat capacity of air"
Pr_a=Prandtl(Air,T=T_a_avg)                              "conductivity of air"
```

The mass flow rate of air passing through the regenerator is evaluated according to the specified ventilation rate:

$$\dot{m}_a = V_{zone}\,ac\,\rho_a$$

The mass flux is evaluated based on the mass flow rate and the area available for the flow; note that only half of the wheel is exposed to each of the two flows and therefore the factor of 2 is required:

$$G = \frac{2\,\dot{m}_a}{A_{fr}\,p}$$

The Reynolds number is evaluated:

$$Re = \frac{4\,G\,r_{char}}{\mu_a}$$

```
m_dot_a=V_zone*ac*rho_a                                  "mass flow rate of air"
```

```
G=2*m_dot_a/(A_fr*p)                                    "mass flux based on open area"
Re=4*G*r_char/mu_a                                      "Reynolds number"
```

The Colburn j_H and friction factors are obtained using the Triangular_channels_ND procedure in EES.

```
call Triangular_channels_ND(Re:f,j_H)   "access convection correlations for triangular channels"
```

The heat transfer coefficient is obtained using the Colburn j_H factor in Eq. (8-388):

$$\overline{h} = \frac{j_H\, G\, c_a}{Pr_a^{2/3}}$$

and used to compute the number of transfer units that characterizes the bed:

$$NTU = \frac{\overline{h}\, A_s}{\dot{m}_a\, c_a}$$

```
h_bar=j_H*G*c_a/Pr_a^(2/3)                              "heat transfer coefficient"
NTU=h_bar*A_s/(m_dot_a*c_a)                             "number of transfer units"
```

The solution for the effectiveness of a balanced, symmetric regenerator (ε) is accessed using the HX procedure in EES. The solution requires the capacitance rate of the air and the equivalent capacitance rate of the matrix, defined as the total heat capacity of the matrix divided by the time that the matrix is in contact with each of the air streams (the blow time, t_B). The matrix material is exposed to each stream for half of a rotation, therefore the blow time is computed according to:

$$t_B = \frac{1}{2\,N}$$

```
time_B=1/(2*N)                          "blow time period"
eff=HX('Regenerator', NTU, m_dot_a*c_a, V_r*c_r*rho_r/time_B, 'epsilon')
                                        "access solution for balanced,symmetric regenerator"
```

The effectiveness is defined as the ratio of the actual to the maximum possible amount of energy transfer to the outdoor air per rotation. On a rate basis, the effectiveness is therefore:

$$\varepsilon = \frac{\dot{q}}{\dot{m}_a\, c_a \left(T_b - T_o\right)}$$

where $\dot{q}$ is the average rate that heat is transferred to the outdoor air.

```
q_dot=eff*m_dot_a*c_a*(T_b-T_o)         "rate of heat transfer"
```

The total amount of money saved in a month related to avoided heating costs (assuming that no ice is formed in the flow passages) is therefore:

$$\text{heating \$ saved} = t_{month}\, \dot{q}\, hc$$

```
heatingsavings=t_month*q_dot*hc          "total avoided heating cost in a month"
```

which leads to a savings of $1161 in January.

The savings is mitigated by the cost of the electricity required to operate the fan. The pressure drop across the regenerator bed is estimated according to Eq. (8-386):

$$\Delta p = \frac{G^2 f\, L}{2\, \rho_a\, r_{char}} \tag{8-395}$$

```
DELTAp=G^2*f*L/(2*rho_a*r_char)          "pressure drop across bed"
```

The fan power is therefore:

$$\dot{w}_{fan} = \frac{2\, \Delta p\, \dot{m}_a}{\rho_a\, \eta_{fan}} \tag{8-396}$$

where the factor of two in Eq. (8-396) is related to the fact that fan power is required to move both the building air and the outdoor air.

```
w_fan=2*DELTAp*m_dot_a/(rho_a*eta_fan)          "fan power consumption"
```

which leads to $\dot{w}_{fan}$ = 2.1 kW. The electrical cost associated with running the fans for a month is:

$$\text{fan \$} = \dot{w}_{fan}\, t_{month}\, ec \tag{8-397}$$

```
fancost=w_fan*t_month*ec          "fan operating cost"
```

The net savings is the heating cost avoided less the fan cost incurred:

net $ saved = heating $ saved - fan $

```
savings=heatingsavings-fancost          "net monthly savings"
```

which leads to a net monthly savings of $995 in January. The significance of this savings can be put into context by estimating the heating cost in January in the absence of the energy recovery wheel:

$$\text{heating \$ no energy recovery} = t_{month}\, \dot{m}_a\, c_a \left(T_b - T_o\right)$$

```
ventilationcost=m_dot_a*c_a*(T_b-T_o)*t_month*hc        "ventilation cost without energy recovery"
```

which leads to $1653. Therefore, the energy recovery wheel can save the zoo about 60% on its heating costs in January.

The analysis can be extended over an entire year. A parametric table is generated that includes the parameters that characterize each month of the year as well as the interesting thermal and economic results of the analysis. (Note that those months where no heating is required are omitted.) The parameters that specify the characteristics of the month are commented out in the equation window:

```
{Month$='Jan'                  "month"
N_days=31 [day]                "number of days in the month"
T_o_C=-8 [C]                   "monthly average outdoor air temperature in C"}
```

and entered (from Table 1) into the parametric table. The parametric table is solved (Figure 3):

1..10	Month$	N_{days} [day]	$T_{o,C}$ [C]	ventilationcost [$]	eff [-]	heatingsavings [$]	fancost [$]	savings [$]
Run 1	Jan	31	-8	1653	0.7023	1161	165.7	994.9
Run 2	Feb	28	-7	1437	0.7029	1010	149.9	860.2
Run 3	Mar	31	-1	1224	0.7067	865.2	167	698.3
Run 4	Apr	30	7	723.4	0.7115	514.7	162.8	351.9
Run 5	May	31	13	398.4	0.7149	284.8	169.1	115.7
Run 6	Jun	30	19	54.52	0.7179	39.14	164.3	-125.2
Run 7	Sep	30	15	274.4	0.7159	196.5	163.9	32.62
Run 8	Oct	31	10	572	0.7133	408.1	168.7	239.4
Run 9	Nov	30	2	1010	0.7085	715.8	162.1	553.8
Run 10	Dec	31	-6	1529	0.7035	1076	166.1	909.8

Figure 3: Parametric table containing the results of a month-by-month analysis of the energy recovery system.

It is possible to access statistics related to each column of the table by right-clicking the column header and selecting properties. One of the statistics is the sum of each of the entries in the column; the net savings over a year is $4631. It is more convenient to have EES automatically sum each column and place the results of this operation in a final row of the table; this is possible using the $SUMROW directive:

```
$SUMROW ON "create a sum row in the table"
```

When the parametric table is solved again, the sum row is placed at the bottom of the table. The energy recovery system will save $4631 annually or 52% of the $8877 heating cost incurred without the system.

The predicted cost savings are optimistic in that the effect of ice build up during freezing conditions is not considered. Regenerators used for ventilation often include a desiccant to provide mass transfer as well as heat transfer. The desiccant transfers water vapor from the stream of higher relative humidity to the stream of lower relativity, which reduces the freeze-up problem. Regenerators that transfer both heat and mass are called enthalpy exchangers.

One method of defrosting a regenerator wheel is to reduce its rotation speed (and therefore increase the blow time) to the point which the effectiveness of the regenerator is significantly reduced; recall from Section 8.10.3 that the effectiveness of a regenerator is reduced if the utilization is increased. The reduction in effectiveness results in a lower rate of heat transfer from the exhaust air to the incoming ventilation air; therefore, the average temperature of the building air that is exhausted to outdoors increases. If the exhaust air temperature is above the freezing point then it will melt the ice that would otherwise form on the matrix.

b) Determine the rotation speed that will result in melting the ice in January.

The characteristics of January are uncommented in the Equation Window:

```
Month$='Jan'                    "month"
N_days=31 [day]                 "number of days in the month"
T_o_C=-8 [C]                    "monthly average outdoor air temperature in C"
```

The exhaust air temperature is calculated using an energy balance:

$$T_{exhaust} = T_b - \frac{q}{\dot{m}_a\, c_a}$$

```
T_exhaust=T_b-q/(m_dot_a*c_a)                  "exhaust air temperature"
T_exhaust_C=converttemp(K,C,T_exhaust)         "in C"
```

which leads to $T_{exhaust}$ = 0.34°C; while this is above freezing, it is probably not sufficiently warm to melt the ice. The coldest temperature of the regenerator matrix surface will be approximately equal to the average of $T_{exhaust}$ and T_o:

$$T_{r,cold} \approx \frac{T_o + T_{exhaust}}{2}$$

```
T_r_cold=(T_exhaust+T_o)/2          "approximate value of the coldest regenerator surface temperature"
T_r_cold_C=converttemp(K,C,T_r_cold) "in C"
```

which leads to $T_{r,cold}$ = -3.8°C. Figure 4 illustrates $T_{r,cold}$ as a function of the rotational speed.



Figure 4: Approximate cold side regenerator surface temperature and net monthly savings for January as a function of the regenerator rotation rate.

Note that the regenerator surface temperature is always greater than the freezing point of water at N = 0.55 rev/min. However, the monthly savings that results when the regenerator is "turned down" to avoid frosting is only about $500, half of the amount that was predicted when frosting is ignored.